

Práctica 3 - Sistema Solar

David Alberto Medina Medina
Dr. Modesto Fernando Castrillon Santana

19 de febrero de 2019

Índice general

1. Introducción	2
2. Método y materiales	3
2.1. Materiales	3
2.2. Método	3
2.2.1. Clase <code>Galaxy</code>	3
2.2.2. Clase <code>Planet</code>	6
2.2.3. Clase <code>Ring</code>	10
2.2.4. Clase <code>BodyLabel</code>	12
2.2.5. Clase <code>PVectorUV</code>	13
2.2.6. Clase <code>Camera</code>	13
2.2.7. Clase <code>Spaceship</code>	14
3. Resultados y discusión	19
4. Conclusiones	20
Referencias	21

Introducción

Processing es un IDE *opensource* que utiliza *Java* como lenguaje de programación. Este proyecto está desarrollado y mantenido por la *Processing Foundation* que sirve como soporte de aprendizaje para instruir a estudiantes de todo el mundo en el mundo de la codificación dentro del contexto de las artes visuales.

El objetivo de este proyecto de laboratorio consiste en utilizar las herramientas de dibujo que nos proporciona el entorno de programación de *Processing* para contruir un sistema solar.

Durante el desarrollo de esta experiencia, aprenderemos a utilizar las diferentes herramientas gráficas que dispone *Processing* para la representación 3D de objetos gracias a las librerías *OpenGL* que este integra en su framework. Se hace uso de las primitivas que facilitan las tareas de traslación y rotación del sistema de referencia que ofrece la librería de *Processing*.

Método y materiales

2.1. Materiales

El desarrollo de este proyecto se ha llevado a cabo utilizando el IDE de desarrollo de aplicaciones *Java* de *JetBrains*, *IntelliJ*, y las siguientes herramientas:

- Texturas de los planetas que conforman el sistema solar (Scope, s.f.)
- Fichero de sonido utilizado para la música de fondo (*DL-Sounds*, s.f.)
- Modelo *OBJ*, fichero *MTL* y materiales de la nave espacial (Nieves, s.f.)
- Librerías propias de *Processing*: Sound, Core, ...

2.2. Método

2.2.1. Clase Galaxy

En primera instancia se declaran las propiedades de clase responsables de la renderización del Sol y sus planetas, la nave espacial y el Universo que contiene a los dos anteriores.

```
1 private PShape universe;  
2 private Planet sun;  
3 private Spaceship spaceship;
```

El método `main()` llama al método estático homónimo de la clase `PApplet` importado de la librería de *Processing*.

```
1 public static void main(String... args) {  
2     PApplet.main("galaxy.Galaxy");  
3 }
```

El método, `settings()`, es un método propio de *Processing* necesario para realizar las configuraciones iniciales de nuestro programa. En concreto, aquí inicializamos el tamaño de la ventana a 1200×1200 píxeles de resolución y seleccionamos el parámetro de renderización `P3D`.

```
1 public void settings() {  
2     size(1200, 1200, P3D);  
3 }
```

La función principal de `setup()` consiste en realizar las configuraciones e instanciaciones iniciales de los planetas –y, cuando proceda, sus anillos–, la nave

espacial y, el Universo. En este método también se instancian las texturas del Universo, los planetas y anillos.

Como puede observarse, el Sistema Solar está conformado por el Sol y una agrupación de satélites que en conjunto conforman los planetas del Sistema –desde Mercurio hasta Neptuno, incluyendo la Luna.

Además, también se inicializa y reproduce el sonido de fondo de la aplicación instanciando al objeto `SoundFile`.

```

1  public void setup() {
2      background(200);
3
4      noStroke();
5
6      SoundFile soundtrack = new SoundFile(this, "res/sounds/
Orbit Beat 130.wav");
7
8      universe = createShape(SPHERE, 1000);
9      PImage universe_mat = loadImage("https://www.
solarsystemscope.com/textures/download/2k_stars_milky_way.jpg");
10 ;
11     universe.setTexture(universe_mat);
12
13     int[] labelRGB = new int[]{150, 150, 150};
14
15     PShape spaceship_model = loadShape("res/Falcon t45 Rescue
ship/Falcon t45 Rescue ship flying.obj");
16     spaceship = new Spaceship(this, spaceship_model, new
PVector(width/2.f, height/2.f, 600.f),
17         1.5f, .2f, labelRGB);
18
19     PImage sun_mat = loadImage("https://www.solarsystemscope.
com/textures/download/2k-sun.jpg");
20     sun = new Planet(this, "Sun", sun_mat, new PVector(0,0,0),
21         110f, .0f,
22         .0f, .0f, .0f, 2.5f, .0f, .0f, labelRGB);
23
24     PImage mercury_mat = loadImage("https://www.
solarsystemscope.com/textures/download/2k-mercury.jpg");
25     Planet mercury = new Planet(this, "Mercury", mercury_mat,
26         new PVector(-140, 0, 0),
27         9.351f, 140.f, .0f, random(360), .25f,
28         .25f, .01f, random(-45, 45), labelRGB);
29
30     PImage venus_mat = loadImage("https://www.solarsystemscope.
com/textures/download/2k-venus-atmosphere.jpg");
31     Planet venus = new Planet(this, "Venus", venus_mat, new
PVector(-200, 0, 0),
32         15.87f, 200.f, .0f, random(360), .25f,
33         .25f, 177.3f, random(-45, 45), labelRGB);
34
35     PImage earth_mat = loadImage("https://www.solarsystemscope.
com/textures/download/2k-earth-daymap.jpg");
36     Planet earth = new Planet(this, "Earth", earth_mat, new
PVector(-260, 0, 0),
37         15.9f, 260.f, .0f, random(360), .25f,
38         .25f, 23.26f, random(-45, 45), labelRGB);
39
40     PImage moon_mat = loadImage("https://www.solarsystemscope.
com/textures/download/2k-moon.jpg");
41     Planet moon = new Planet(this, "Moon", moon_mat, new
PVector(-40, 0, 0),

```

```

39         6.25f, 40.f, .0f, random(360), .25f,
40         .25f, 1.5f, random(-45, 45), labelRGB);
41
42         PImage mars_mat = loadImage("https://www.solarsystemscope.
43 com/textures/download/2k-mars.jpg");
44         Planet mars = new Planet(this, "Mars", mars_mat, new
45 PVector(-320, 0, 0),
46         15.487f, 320.f, .0f, random(360), .25f,
47         .25f, 25.19f, random(-45, 45), labelRGB);
48
49         PImage jupiter_mat = loadImage("https://www.
50 solarsystemscope.com/textures/download/2k-jupiter.jpg");
51         Planet jupiter = new Planet(this, "Jupiter", jupiter_mat,
52 new PVector(-380, 0, 0),
53         30.1f, 380.f, .0f, random(360), .25f,
54         .25f, 3.13f, random(-45, 45), labelRGB);
55
56         PImage saturn_mat = loadImage("https://www.solarsystemscope
57 .com/textures/download/2k-saturn.jpg");
58         Planet saturn = new Planet(this, "Saturn", saturn_mat, new
59 PVector(-440, 0, 0),
60         25.07f, 440.f, .0f, random(360), .25f,
61         .25f, 26.73f, random(-45, 45), labelRGB);
62
63         PImage saturn_ring_mat = loadImage("https://www.
64 solarsystemscope.com/textures/download/2k-saturn-ring-alpha.png
65 ");
66         Ring saturn_ring = new Ring(this, new PVector(0, 0, 0),
67 saturn.getRadius() + saturn.getRadius() * 1.2f,
68         saturn.getRadius() + saturn.getRadius() * .025f,
69 saturn_ring_mat, 200);
70
71         PImage uranus_mat = loadImage("https://www.solarsystemscope
72 .com/textures/download/2k-uranus.jpg");
73         Planet uranus = new Planet(this, "Uranus", uranus_mat, new
74 PVector(-500, 0, 0),
75         10.92f, 500.f, .0f, random(360), .25f,
76         .25f, 97.77f, random(-45, 45), labelRGB);
77
78         PImage neptune_mat = loadImage("https://www.
79 solarsystemscope.com/textures/download/2k-neptune.jpg");
80         Planet neptune = new Planet(this, "Neptune", neptune_mat,
81 new PVector(-560, 0, 0),
82         10.36f, 560.f, .0f, random(360), .25f,
83         .25f, 28.32f, random(-45, 45), labelRGB);
84
85         earth.addSatellite(moon);
86
87         saturn.setRing(saturn_ring);
88
89         sun.addSatellite(mercury);
90         sun.addSatellite(venus);
91         sun.addSatellite(earth);
92         sun.addSatellite(mars);
93         sun.addSatellite(jupiter);
94         sun.addSatellite(saturn);
95         sun.addSatellite(uranus);
96         sun.addSatellite(neptune);
97
98         soundtrack.play();
99     }

```

El método `draw()` es el bucle principal del programa en donde se refresca el estado de la nave espacial, el Universo y los planetas. Como puede observarse, el sistema de referencia se traslada al centro de la pantalla.

```

1  public void draw() {
2      spaceship.refresh();
3
4      translate(width/2.f, height/2.f, 0);
5
6      shape(universe);
7
8      sun.refresh();
9  }
```

El usuario puede mover la nave espacial que se genera en el mundo pulsando los botones A, W, S y D, y controlar la velocidad de la nave al presionar + y , para acelerar o desacelerar la nave –o incluso ir marcha atrás.

```

1  public void keyTyped() {
2      if (key == '+') {
3          spaceship.updateSpeed(spaceship.POWERUP);
4      } else if (key == '-') {
5          spaceship.updateSpeed(spaceship.POWERDOWN);
6      } else if (key == 's') {
7          spaceship.updatePitch(spaceship.UP);
8      } else if (key == 'w') {
9          spaceship.updatePitch(spaceship.DOWN);
10     } else if (key == 'a') {
11         spaceship.updateRoll(spaceship.LEFT);
12     } else if (key == 'd') {
13         spaceship.updateRoll(spaceship.RIGHT);
14     }
15 }
```

2.2.2. Clase Planet

La renderización de los planetas y satélites del Sistema Solar son gestionados en esta clase. Es por este motivo que son necesarios los siguientes parámetros para instanciar un objeto de la clase **Planet**:

parent Objeto de la clase **PApplet** que nos permitirá llamar a las primitivas de dibujo de *Processing*

name String que identifica al planeta en cuestión. Esta variable será utilizada para el etiquetado del planeta.

material Objeto de la clase **PImage** que contiene las texturas del planeta.

pos Objeto de la clase **PVector** que define la posición inicial del planeta.

radius Número en coma flotante que define el radio del planeta.

orbitRadius Número en coma flotante que define el radio de la órbita de traslación del planeta.

rotationDegree Número en coma flotante que define el ángulo de rotación del planeta.

translationDegree Número en coma flotante que define el ángulo de traslación del planeta en su órbita.

translationSpeed Número en coma flotante que define la velocidad de traslación.

rotationSpeed Número en coma flotante que define la velocidad de rotación del planeta.

axialTilt Número en coma flotante que define el ángulo del eje de rotación del planeta. Este parámetro define los ángulos reales de los ejes de traslación de cada planeta del Sistem Solar (Astronoo, s.f.).

orbitTilt Número en coma flotante que define el ángulo que describe la órbita del planeta con la horizontal.

labelRGB Array de enteros que define los valores RGB del color de la etiqueta del planeta.

```
1  private final float TRANSLATION_SPEED;
2  private final float ROTATION_SPEED;
3  private final float AXIAL_TILT;
4  private final float ORBIT_TILT;
5  private final PImage MATERIAL;
6
7  private final String PLANET_LABEL;
8
9  private PApplet parent;
10 private PShape planet;
11 private Ring ring;
12 private List<Planet> moons;
13 private int[] labelRGB;
14
15 private PVector pos;
16 private float radius;
17 private float orbitRadius;
18 private float rotationDegree;
19 private float translationDegree;
20
21 public Planet(PApplet parent, String name, PImage material,
22 PVector pos, float radius, float orbitRadius, float
23 rotationDegree, float translationDegree, float translationSpeed,
24 float rotationSpeed, float axialTilt, float orbitTilt, int[]
25 labelRGB) {
26     TRANSLATION_SPEED = PApplet.radians(translationSpeed);
27     ROTATION_SPEED = PApplet.radians(rotationSpeed);
28     AXIAL_TILT = PApplet.radians(axialTilt);
29     ORBIT_TILT = PApplet.radians(orbitTilt);
30
31     PLANET_LABEL = name;
32
33     this.pos = pos;
34     this.radius = radius;
35     this.orbitRadius = orbitRadius;
36     this.MATERIAL = material;
37     this.rotationDegree = PApplet.radians(rotationDegree);
38     this.translationDegree = PApplet.radians(translationDegree);
39
40     ;
41     this.parent = parent;
```



```

37         moons = new ArrayList<>();
38         planet = parent.createShape(PConstants.SPHERE, radius);
39         planet.setTexture(material);
40         this.labelRGB = labelRGB;
41     }
42

```

Es posible añadir satélites al planeta –método `addSatellite()`–. Además, un planeta puede devolver la lista de satélites que ya tiene almacenado –método `getSatellites()`– y el valor de su radio –método `getRadius()`.

```

1     public void addSatellite(Planet planet) {
2         moons.add(planet);
3     }
4
5     public List<Planet> getSatellites() {
6         return moons;
7     }
8
9     public float getRadius() {
10        return radius;
11    }

```

El planeta Saturno se caracteriza por tener un anillo de asteroides que lo rodea. Es por ello, que la clase `Planet` dispone de un método para generar dicho anillo, el método `setRing()`. También se puede comprobar si el planeta anillo –`hasRing()`–.

```

1     public void setRing(Ring ring) {
2         this.ring = ring;
3     }
4
5     private boolean hasRing() {
6         if(ring != null)
7             return true;
8         else
9             return false;

```

Un planeta dispone de una etiqueta que lo identifica con su nombre. Con el objetivo de que las etiquetas estén bien alineadas se deshacen las rotaciones de la órbita y radio del planeta que hayan podido tener lugar cuando se llama al método `refresh()` –ver más adelante. A continuación, se instancia un objeto de la clase `BodyLabel` que es responsable del etiquetado de los planetas.

```

1     private void label() {
2         parent.rotateY(-translationDegree);
3         parent.rotateZ(-ORBIT_TILT);
4
5         BodyLabel bl = new BodyLabel(parent, PLANET_LABEL, radius +
6         100, 1, labelRGB);
7         bl.refresh();
8     }

```

Una vez todos que los parámetros del planeta han sido correctamente configurados se procede con el proceso de visualización del mismo, el cual tiene lugar dentro del método `refresh()` siguiendo una serie de pasos:

En primer lugar, se guarda el contexto de la matriz de transformación del sistema de coordenadas actual llamando a la primitiva `popMatrix()`. A continuación, se realiza una rotación del sistema de referencia sobre el eje Z e Y para ajustar los ejes de rotación y traslación del planeta.

Una vez concluido este paso, se actualiza el valor del ángulo de traslación –restringiéndose a un máximo de 2π radianes– y se procede a trasladar el eje de coordenadas tantas unidades como indica el parámetro del radio de órbita.

En segundo lugar, se guarda el contexto de la matriz de transformación anterior para poder ajustar el ángulo de rotación y el ángulo del eje de rotación del planeta. Se comprueba si el planeta tiene anillos –si es así se procede con su rederizado–, se procede con el proceso de visualización del planeta y se sale del contexto creado llamando a la primitiva `popMatrix()`.

En tercer lugar, se comprueba si el planeta tiene algún satélite y si es así se abre un contexto nuevo para el mismo llamando a las primitivas `pushMatrix()` y `popMatrix()` y se deshace, ordenadamente, la transformación de rotación del sistema de referencia que tuvo lugar en el primer paso. A continuación, se llama al método `refresh()` del satélite en cuestión.

Por último, se llama al método `label()` para generar la visualización de las etiquetas de los planetas y se sale del contexto –`popMatrix()`– generado al principio del método.

```

1      public void refresh() {
2          parent.pushMatrix();
3
4          parent.rotateZ(ORBIT_TILT);
5          parent.rotateY(translationDegree);
6
7          translationDegree = (translationDegree + TRANSLATION_SPEED)
8          % PApplet.radians(360);
9
10         parent.translate(orbitRadius, 0, 0);
11
12
13         parent.pushMatrix();
14         parent.rotateY(rotationDegree);
15         rotationDegree = (rotationDegree + ROTATION_SPEED) %
16         PApplet.radians(360);
17
18         parent.rotateZ(AXIAL_TILT);
19
20         if(hasRing()) ring.refresh();
21
22         parent.shape(planet);
23
24         if (hasRing()) ring.refresh();
25
26         parent.popMatrix();
27
28         if (getSatellites() != null && getSatellites().size() > 0)
29         {
30             for (Planet satellite : getSatellites()) {
31                 parent.pushMatrix();
32                 parent.rotateY(-translationDegree);
33                 parent.rotateZ(-ORBIT_TILT);
34                 satellite.refresh();
35                 parent.popMatrix();
36             }
37         }
38         label();
39         parent.popMatrix();

```

2.2.3. Clase Ring

Esta clase es la responsable de la visualización de los anillos del planeta Saturno. Como parámetros de inicialización tenemos:

parent Objeto de la clase **PApplet** que nos permitirá llamar a las primitivas de dibujo de *Processing*.

pos Objeto de la clase **PVector** que define la posición inicial del anillo.

outerRadio Número en coma flotante que define el radio externo del anillo.

innerRing Número en coma flotante que define el radio interno del anillo.

material Objeto de la clase **PImage** que contiene las texturas del anillo.

numParts Número entero que define el número de partes de las que está hecha el anillo.

```
1  private final int NUMPARTS;
2  private final PImage MATERIAL;
3
4  private PShape ring;
5
6  private PApplet parent;
7  private List<PVectorUV> outerRing;
8  private List<PVectorUV> innerRing;
9
10 private PVector pos;
11 private float outerRadius;
12 private float innerRadius;
13
14 public Ring(PApplet parent, PVector pos, float outerRadius,
15 float innerRadius, PImage material, int numParts) {
16     this.NUMPARTS = numParts;
17
18     this.parent = parent;
19     this.pos = pos;
20     this.outerRadius = outerRadius;
21     this.innerRadius = innerRadius;
22     this.MATERIAL = material;
23
24     this.outerRing = new ArrayList<>();
25     this.innerRing = new ArrayList<>();
26
27     renderRing();
28 }
```

El constructor de la clase **Ring** llama al método privado **renderRing()** que es el responsable de generar el anillo.

```
1  private void renderRing() {
2      float angleStep = PApplet.radians(360.f/NUMPARTS);
3
4      float angle = .0f;
5
6      for (int i = 0; i <= NUMPARTS; i++) {
7          angle = (i * angleStep) % PApplet.radians(360);
8          if ((i % 2) == 0) {
9              outerRing.add(buildRingPerimeter(angle, outerRadius
, true, false));
10          }
```

```

10         innerRing.add(buildRingPerimeter(angle, innerRadius
11         , false, false));
12     } else {
13         outerRing.add(buildRingPerimeter(angle, outerRadius
14         , true, true));
15         innerRing.add(buildRingPerimeter(angle, innerRadius
16         , false, true));
17     }
18 }

```

Este método llama al método privado `buildRingPerimeter()` que crea los vértices que conforman el contorno interno y externo del anillo. Es en este método, donde se definen además los componentes de los vectores u y v , de modo que se pueda mapear correctamente la imagen con la texturas con una celda que conformará parte del anillo del planeta.

```

1 private PVectorUV buildRingPerimeter(float angle, float radius,
2 boolean isOuter, boolean isNextPair) {
3     float x = radius * PApplet.cos(angle) - radius * PApplet.
4     sin(angle);
5     float y = 0;
6     float z = radius * PApplet.sin(angle) + radius * PApplet.
7     cos(angle);
8     if(!isNextPair)
9         if(isOuter)
10             return new PVectorUV(x, y, z, 0, MATERIAL.height);
11         else
12             return new PVectorUV(x, y, z, MATERIAL.width,
13             MATERIAL.height);
14     else
15         if(isOuter)
16             return new PVectorUV(x, y, z, 0, 0);
17         else
18             return new PVectorUV(x, y, z, MATERIAL.width, 0);
19 }

```

Finalmente, la visualización del anillo tiene lugar en el método `refresh()` donde se creará el anillo y se asignará su textura.

```

1 public void refresh() {
2     ring = parent.createShape();
3
4     ring.beginShape(PConstants.QUAD_STRIP);
5     ring.noStroke();
6     ring.texture(MATERIAL);
7
8     setUpVertices();
9
10    ring.endShape(PConstants.CLOSE);
11    parent.shape(ring);
12 }

```

Para que la asignación de textura tenga efecto, se selecciona la marca `QUAD_STRIP` para que, una vez definidos los vértices de manera sistemática, se generen rectángulos contiguos sobre los cuales se aplicará de manera ordenada las texturas.

```

1 private void setUpVertices() {
2     for (int i = 0; i < NUMPARTS; i++) {
3         PVectorUV outer1 = outerRing.get(i % NUMPARTS);

```

```

4         PVectorUV inner1 = innerRing.get(i % NUMPARTS);
5         assignVerticesSet(outer1, inner1);
6
7         PVectorUV outer2 = outerRing.get((i+1) % NUMPARTS);
8         PVectorUV inner2 = innerRing.get((i+1) % NUMPARTS);
9         assignVerticesSet(outer2, inner2);
10    }
11 }
12
13 private void assignVerticesSet(PVectorUV outer, PVectorUV inner
14 ) {
15     ring.vertex(outer.x, outer.y, outer.z, outer.u, outer.v);
16     ring.vertex(inner.x, inner.y, inner.z, inner.u, inner.v);
17 }

```

2.2.4. Clase BodyLabel

Esta clase forma parte del paquete `galaxy.utils` y permite la visualización de las etiquetas de los planetas. Para instanciar un objeto de esta clase necesitaremos pasarle los siguientes parámetros:

parent Objeto de la clase `PApplet` que nos permitirá llamar a las primitivas de dibujo de *Processing*.

label Variable de tipo `String` que contiene el nombre del planeta a etiquetar.

lineLength Número en coma flotante que define la longitud de la línea que apunta hacia el planeta a etiquetar.

lineWeight Numero en coma flotante que define el grosor de la línea que apunta al planeta a etiquetar.

rgb Array de tipo entero que contiene los valores RGB del color de la etiqueta.

```

1     private final String LABELTXT;
2
3     private final PApplet parent;
4
5     private PMatrix pos;
6     private float lineLength;
7     private int lineWeight;
8     private int[] rgb;
9
10    public BodyLabel(PApplet parent, String label, float lineLength
11    , int lineWeight, int[] rgb) {
12        this.LABELTXT = label;
13
14        this.parent = parent;
15        this.pos = pos;
16        this.lineLength = lineLength;
17        this.lineWeight = lineWeight;
18        this.rgb = rgb;
19    }

```

El método `refresh()` muestra por pantalla la etiqueta con el nombre del planeta y la flecha que apunta hacia el mismo.

```

1  public void refresh() {
2      parent.pushMatrix();
3
4      parent.setMatrix(pos);
5      parent.stroke(rgb[0], rgb[1], rgb[2]);
6      parent.strokeWeight(lineWeight);
7      parent.fill(rgb[0], rgb[1], rgb[2]);
8      parent.line(0,0,0,0,-lineLenght,0);
9      parent.text(LABELTXT,5,-lineLenght+10,0);
10
11     parent.popMatrix();
12 }

```

2.2.5. Clase PVectorUV

Se trata de una clase que extiende de la clase **PVector** que se utiliza para almacenar de manera ordenada las coordenadas de un vértice y las coordenadas UV de su vector de mapeo correspondiente. Esta clase es usada por la clase definida en la sección 2.2.3.

```

1  public class BodyLabel extends PApplet {
2      private final String LABELTXT;
3
4      private final PApplet parent;
5
6      private PMatrix pos;
7      private float lineLenght;
8      private int lineWeight;
9      private int[] rgb;

```

2.2.6. Clase Camera

Esta clase pertenece al paquete **camera** y gestiona todas las operaciones que se realicen sobre la cámara. Un objeto de la clase **Camera** toma como parámetros de inicialización:

parent Objeto de la clase **PApplet** que nos permitirá llamar a las primitivas de dibujo de *Processing*.

pos Objeto de la clase **PVector** que define la posición inicial de la cámara.

center Objeto de la clase **PVector** que define la posición inicial del centro o foco de la cámara.

rotation Objeto de la clase **PVector** que define la rotación inicial de la cámara.

```

1  private PApplet parent;
2  private PVector pos;
3  private PVector center;
4  private PVector rotation;
5
6  public Camera(PApplet parent, PVector pos, PVector center,
7  PVector rotation) {
8      this.parent = parent;
9      this.pos = pos;
10     this.center = center;

```

```

10         this.rotation = rotation;
11     }
12
13     public Camera(PApplet parent, PVector pos, PVector center) {
14         this.parent = parent;
15         this.pos = pos;
16         this.center = center;
17         this.rotation = new PVector(0,1,0);
18     }
19
20     public Camera(PApplet parent) {
21         this.parent = parent;
22         reset();
23     }

```

La inicialización o *reset* de los parámetros de la cámara a sus valores por defecto se establecen llamando al método **reset()** (ver (Processing, s.f.)).

```

1     public void reset() {
2         this.pos = new PVector(parent.width/2.f, parent.height/2.f,
3         (parent.height/2.f) / PApplet.tan(PApplet.PI*30.f / 180.f));
4         this.center = new PVector(parent.width/2.f, parent.height
5         /2.f, .0f);
6         this.rotation = new PVector(.0f,1.f,.0f);
7     }

```

La posición, centro y rotación de la cámara pueden ser fácilmente modificados llamando a sus correspondientes *setters*.

```

1     public void setPosition(float x, float y, float z) {
2         pos.x = x;
3         pos.y = y;
4         pos.z = z;
5     }
6
7     public void setCenter(float x, float y, float z) {
8         center.x = x;
9         center.y = y;
10        center.z = z;
11    }
12
13    public void setRotation(float x, float y, float z) {
14        rotation.x = x;
15        rotation.y = y;
16        rotation.z = z;
17    }

```

El método **refresh()** aplica los cambios en la cámara, llamando a la primitiva **camera()** de *Processing*.

```

1     public void refresh() {
2         parent.camera(pos.x, pos.y, pos.z, center.x, center.y,
3         center.z, rotation.x, rotation.y, rotation.z);
4     }

```

2.2.7. Clase Spaceship

Esta clase es la responsable de la configuración y visualización de la nave espacial. Sus parámetros de inicialización son:

parent Objeto de la clase **PApplet** que nos permitirá llamar a las primitivas de dibujo de *Processing*.

spaceship Objeto de la clase **PShape** que contiene el modelo en 3D –archivo OBJ-- de la nave espacial.

pos Objeto de la clase **PVector** que define la posición inicial de la nave espacial.

speed Número en coma flotante que define la velocidad con la que se puede mover la nave.

scale Número en coma flotante que define el escalado de la nave espacial.

labelRGB Array de tipo entero que contiene los valores RGB del color del texto de inicio de juego.

Otras variables importantes son:

headingX Número en coma flotante que define el ángulo de la nave con respecto al eje X.

headingY Número en coma flotante que define el ángulo de la nave con respecto al eje Y.

headingZ Número en coma flotante que define el ángulo de la nave con respecto al eje Z.

pitch Número en coma flotante que define el ángulo de *cabeceo* acumulado.

roll Número en coma flotante que define el ángulo de *alabeo* acumulado.

```
1  private final static float POS_X.SPAN = 5.f;
2  private final static float POS_Y.SPAN = -50.f;
3  private final static float POS_Z.SPAN = 200.f;
4  private final static float CENTER_DISTANCE = -200.f;
5  private final static float MAX_SPEED = 10.f;
6  private final static float SPEED_STEP = .25f;
7  private final static float PITCH_STEP = PApplet.radians(.5f);
8  private final static float ROLL_STEP = PApplet.radians(1.f);
9  private final static float ROTATION_ANIMATION = PApplet.radians
(200.f);
10
11 public final static float UP = -1.f;
12 public final static float DOWN = 1.f;
13 public final static float LEFT = -1.f;
14 public final static float RIGHT = 1.f;
15 public final static float POWERDOWN = -1.f;
16 public final static float POWERUP = 1.f;
17
18
19 private PApplet parent;
20 private PShape spaceship;
21 private PVector pos;
22 private float speed;
23 private Camera cam;
24
25 private float headingX;
26 private float headingY;
27 private float headingZ;
```



```

28
29     private float pitch;
30     private float roll;
31
32     private int intro;
33     private int[] labelRGB;
34
35
36     public Spaceship(PApplet parent, PShape spaceship, PVector pos,
37         float speed, float scale, int[] labelRGB) {
38         this.parent = parent;
39         this.spaceship = spaceship;
40         this.pos = pos;
41         this.speed = speed;
42         this.cam = new Camera(parent);
43
44         headingX = PApplet.cos(-90);
45         headingY = PApplet.cos(90);
46         headingZ = PApplet.cos(180);
47
48         pitch = .0f;
49         roll = .0f;
50
51         intro = 0;
52         this.labelRGB = labelRGB;
53
54         this.spaceship.scale(scale);
55         this.spaceship.rotateX(PApplet.radians(180));
56         this.spaceship.rotateY(PApplet.radians(3));
57         setupCameraView();
58     }

```

Al inicializar un objeto de la clase `Spaceship`, se ajusta la posición de la cámara.

```

1     }
2
3     public void refresh() {
4         moveForward();
5
6         cam.setPosition(pos.x + POS_X_SPAN, pos.y + POS_Y_SPAN, pos
7             .z + POS_Z_SPAN);
8         cam.setCenter(pos.x, pos.y, pos.z + CENTER_DISTANCE);
9         cam.refresh();
10
11         parent.pushMatrix();
12         parent.translate(pos.x, pos.y, pos.z);
13
14         intro();
15
16         parent.rotateX(pitch * ROTATION_ANIMATION);
17         parent.rotateZ(roll * ROTATION_ANIMATION);
18         parent.shape(spaceship);
19         parent.popMatrix();
20     }
21
22     private void setupCameraView() {
23         cam.setPosition(pos.x + POS_X_SPAN, pos.y + POS_Y_SPAN, pos
24             .z + POS_Z_SPAN);
25         cam.setCenter(pos.x, pos.y, pos.z + CENTER_DISTANCE);
26         cam.setRotation(.0f, 1.f, 0.f);
27         cam.refresh();
28     }

```

La velocidad y, los ángulos de cabeceo y/o alabeo son actualizados cada vez que el usuario interactúa con la aplicación a través del teclado, llamándolos a los métodos `updateSpeed()`, `updatePitch()` y `updateRoll()`, respectivamente.

Cada vez que se actualiza el ángulo de cabeceo, se suma una cantidad discreta de ángulo definida en una constante de clase `PITCH_STEP` y `ROLL_STEP` en las variables `pitch` y `roll`. Este ángulo de cabeceo y alabeo serán utilizados para definir la nueva ruta de movimiento de la nave espacial con respecto al eje Y `-pitch` o ángulo cabeceo y eje X `-roll` o ángulo de alabeo.

```

1  public void updateSpeed(float speedDir) {
2      this.speed += speedDir * SPEED_STEP % MAX_SPEED;
3  }
4
5  public void updatePitch(float pitchDir) {
6      pitch += pitchDir * PITCH_STEP % PApplet.radians(360);
7      headingY += pitchDir * PApplet.cos(pitch);
8  }
9
10 public void updateRoll(float rollDir) {
11     roll += rollDir * ROLL_STEP % PApplet.radians(360);
12     headingX += rollDir * PApplet.cos(roll);
13 }

```

El movimiento de la nave está definido por sus coordenadas polares, tal que

$$\left. \begin{aligned} x &= speed \times \cos_x \alpha \\ y &= speed \times \cos_y \beta \\ z &= speed \times \cos_z \gamma \end{aligned} \right\}$$

donde $\cos_x \alpha$, $\cos_y \beta$ y $\cos_z \gamma$ corresponde al *heading* con respecto al eje X, Y y Z, respectivamente.

```

1  private void moveForward() {
2      pos.x += speed * headingX;
3      pos.y += speed * headingY;
4      pos.z += speed * headingZ;
5  }

```

Existe también una función para mostrar un pequeño tutorial que indica al usuario los controles necesarios para controlar la nave.

```

1  private void intro() {
2      parent.stroke(labelRGB[0], labelRGB[1], labelRGB[2]);
3      if(intro < 40) parent.text("PULSE A, W, S o D para moverse",
4      , -150, -150, -200);
5      else if (intro > 40 && intro < 80) parent.text("PULSE + o -
6      para acelerar y desacelerar", -150, -150, -200);
7      if (intro < 80) intro++;
8  }

```

La cámara se actualiza junto con la posición de la nave espacial a lo largo de su viaje por el espacio. Los movimientos de cabeceo y alabeo generan una rotación de la nave que puede ser fácilmente apreciable por pantalla.

```

1  public void refresh() {
2      moveForward();
3  }

```

```

4      cam.setPosition(pos.x + POS_X_SPAN, pos.y + POS_Y_SPAN, pos
      .z + POS_Z_SPAN);
5      cam.setCenter(pos.x, pos.y, pos.z + CENTER_DISTANCE);
6      cam.refresh();
7
8      parent.pushMatrix();
9      parent.translate(pos.x, pos.y, pos.z);
10
11     intro();
12
13     parent.rotateX(pitch * ROTATION_ANIMATION);
14     parent.rotateZ(roll * ROTATION_ANIMATION);
15     parent.shape(spaceship);
16     parent.popMatrix();
17 }

```

Resultados y discusión

Los materiales software y métodos aquí descritos han sido favorables para la realización y culminación del proyecto. Se ha conseguido diseñar y visualizar un sistema planetario ((Medina, s.f.)). Las traslaciones y rotaciones de los planetas se han implementado con relativa sencillez gracias a las primitivas que dispone *Processing* para manipular el sistema de coordenadas.

Se ha logrado trasladar y establecer un método de seguimiento de la nave espacial a lo largo de su trayectoria por el espacio. Ha sido posible migrar del IDE oficial de *Processing* a *IntelliJ*.




Figura 3.1: Nave espacial viajando por el Sistema Solar

Conclusiones

El conocimiento de las coordenadas polares ha sido de real importancia a la hora de definir un método para establecer el movimiento de la nave.

Las primitivas de *Processing* para la manipulación del sistema de referencia han facilitado en gran medida el proceso de generación y movimiento de los planetas y satélites del Sistema Solar. Los métodos de *Processing* de gestión de la pila de las matrices de transformación –`pushMatrix()` y `popMatrix()`– han sido de gran importancia para establecer contextos sobre los cuales se definen de manera aislada transformaciones lineales sobre el sistema de referencia.

Referencias

- Astronoo. (s.f.). *Inclination of the Planets*. <http://www.astronoo.com/en/articles/axial-tilt-planets.html>. Autor. (Accessed: 2019-02-19)
- DL-Sounds. (s.f.). <https://www.dl-sounds.com/royalty-free/orbit-beat-130/>. (Accessed: 2019-02-19)
- Medina, D. A. M. (s.f.). *Github: Solar System*. <https://github.com/david00medina/solar-system>. Github. (Accessed: 2019-02-19)
- Nieves, H. (s.f.). *Falcon t45 Rescue ship*. <https://www.turbosquid.com>. Turbo Squid. (Accessed: 2019-02-19)
- Processing. (s.f.). *Processing: Reference*. <https://processing.org/reference/>. Autor. (Accessed: 2019-02-19)
- Scope, S. S. (s.f.). *Solar Textures*. <https://www.solarsystemscope.com/textures/>. Autor. (Accessed: 2019-02-19)